
7-5-S-Laplace Transform Issues and Opportunities in *Mathematica*

The Laplace Transform is a mathematical construct that has proven very useful in both solving and understanding differential equations. We introduce it and show its power here.

Definition of Laplace Transform of a function $f(t)$ if the integral converges:

$$\mathcal{L}(f(t))(s) = F(s) = \int_0^{\infty} e^{-st} f(t) dt$$

In order to evaluate such an integral we might be using integration by parts for $f(t)$ of a different "sort" than exponential functions. However, we turn to *Mathematica* to do the integrations for us and use its own powerful computation, look-up, and pattern matching algorithms to do Laplace Transforms.

This "strange" integral transforms most functions (if the integral converges) from a t domain to an s domain where s is (in our case) a real positive number.

```
Integrate[Exp[- s t] f[t], {t, 0, Infinity}]
```

```
ConditionalExpression[ $\frac{1}{s}$ , Re[s] > 0]
```

```
Integrate[Exp[- s t] Cos[3 t], {t, 0, Infinity}]
```

```
ConditionalExpression[ $\frac{s}{9 + s^2}$ , Re[s] > 0]
```

Notice the restriction that $\text{Re}[s] > 0$ in order for us to have convergence to the quantity, $s/(9+s^2)$. Thus $s/(9+s^2)$ IS the Laplace Transform of the function $\cos(3t)$.

Tables of these exist and we could visit them, however since *Mathematica* can do these instantly without us looking them up we do our computations of Laplace Transforms in *Mathematica*.

Often the s domain is called the frequency domain, and the t domain is called the time domain, for if the following integral - the Laplace Transform - is to make sense with respect to units then the units of s must be $1/\text{Time}$, for in this way $s*t$ has no units and so $\text{Exp}[-s*t]$ has no units. Thus the Laplace Transform integral is obtained by just integrating a function of t over an interval of time, $[0, \infty)$ - resulting in a function of s , we call it $\mathcal{L}(f(t))(s) = F(s)$.

```
F[s_] = Integrate[Exp[-s t] f[t], {t, 0, Infinity}]
```

$$\int_0^{\infty} e^{-s t} f[t] dt$$

Mathematica can compute these Laplace Transforms and we see how to do this. Here are the definitions brought up by query from *Mathematica*. If one needs more, such as an example, then go to Help Menu and type "LaplaceTransform" and you will have a window offering up a definition with an example worked out at the bottom. OR just type one of the following

?? LaplaceTransform

LaplaceTransform[*expr*, *t*, *s*] gives the Laplace transform of *expr*.

LaplaceTransform[*expr*, {*t*₁, *t*₂, ...}, {*s*₁, *s*₂, ...}] gives the multidimensional Laplace transform of *expr*. >>

```
Attributes[LaplaceTransform] = {Protected, ReadProtected}
```

```
Options[LaplaceTransform] = {Assumptions -> $Assumptions,
  GenerateConditions -> False, PrincipalValue -> False, Analytic -> True}
```

We have a tougher trick in undoing the Laplace Transform and this is done using the InverseLaplaceTransform command.

?? InverseLaplaceTransform

InverseLaplaceTransform[*expr*, *s*, *t*] gives the inverse Laplace transform of *expr*.

InverseLaplaceTransform[*expr*, {*s*₁, *s*₂, ...}, {*t*₁, *t*₂, ...}] gives the multidimensional inverse Laplace transform of *expr*. >>

```
Attributes[InverseLaplaceTransform] = {Protected, ReadProtected}
```

So here we go . . . taking Laplace Transforms of functions We Build Up a Table of Laplace Transforms and attempt to ascertain patterns.

`LaplaceTransform[1, t, s]`

$$\frac{1}{s}$$

`LaplaceTransform[t, t, s]`

$$\frac{1}{s^2}$$

`LaplaceTransform[t^2, t, s]`

$$\frac{2}{s^3}$$

`LaplaceTransform[t^3, t, s]`

$$\frac{6}{s^4}$$

`LaplaceTransform[t^4, t, s]`

$$\frac{24}{s^5}$$

What do you believe we might get for the Laplace Transform of t^n ? Of course, we could stop and offer a formal induction proof of general formula, but we proceed to take the Laplace Transform of other classes of functions.

Trig functions

`LaplaceTransform[Sin[3 t], t, s]`

$$\frac{3}{9 + s^2}$$

`LaplaceTransform[Sin[5 t], t, s]`

$$\frac{5}{25 + s^2}$$

`LaplaceTransform[Cos[3 t], t, s]`

$$\frac{s}{9 + s^2}$$

`LaplaceTransform[Cos[5 t], t, s]`

$$\frac{s}{25 + s^2}$$

See a pattern here?

... and exponential functions.

`LaplaceTransform[Exp[3 t], t, s]`

$$\frac{1}{-3 + s}$$

`LaplaceTransform[Exp[5 t], t, s]`

$$\frac{1}{-5 + s}$$

`LaplaceTransform[Exp[-7 t], t, s]`

$$\frac{1}{7 + s}$$

Incidentally, the transform of functions like $f(t) = e^{at}$ and $f(t) = k$ are easy by hand efforts! Try one.

Continuing in this manner of discovery we could fill in a typical Laplace Transform Table. A sample, but short and simple form, table is the following:

$f(t)$	$L(f)(t)$
1	$\frac{1}{s}, s > 0$
t^n	$\frac{n!}{s^{n+1}}, s > 0$
$\sin(at)$	$\frac{a}{s^2 + a^2}, s > 0$
$\cos(at)$	$\frac{s}{s^2 + a^2}, s > 0$
e^{at}	$\frac{1}{s - a}, s > a$
$e^{at} \sin(bt)$	$\frac{b}{(s - a)^2 + b^2}, s > a$
$e^{at} \cos(bt)$	$\frac{s - a}{(s - a)^2 + b^2}, s > a$
$t^n e^{at}$	$\frac{n!}{(s - a)^{n+1}}, s > a$

We could combine trig and exponential functions . . .

`LaplaceTransform[Exp[5 t] Sin[3 t], t, s]`

$$\frac{3}{9 + (-5 + s)^2}$$

... and in general to match the table entry.

```
LaplaceTransform[Exp[a t] Sin[b t], t, s]
```

$$\frac{b}{b^2 + (a - s)^2}$$

If we are given the Laplace Transform $Y(s)$ how might we obtain the original, i.e. the $y(t)$ whose Laplace Transform is $Y(s)$. We try it on some of our observed Transforms and some others functions.

```
InverseLaplaceTransform[1 / (s - 4), s, t]
```

$$e^{4t}$$

```
InverseLaplaceTransform[s - 3 / ((s - 3)^2 + 4), s, t]
```

$$\frac{3}{4} i e^{(3-2i)t} (-1 + e^{4it}) + \text{DiracDelta}'[t]$$

```
InverseLaplaceTransform[1 / s^7, s, t]
```

$$\frac{t^6}{720}$$

```
InverseLaplaceTransform[s / (s^2 + 4) + 1 / (s - 4), s, t]
```

$$e^{4t} + \text{Cos}[2t]$$


```
InverseLaplaceTransform[s / (s^2 - 4), s, t]
```

$$\frac{1}{2} e^{-2t} (1 + e^{4t})$$

```
ysol[t_] = InverseLaplaceTransform[(s + 3) / ((s + 3)^2 + 4), s, t]
```

$$\frac{1}{2} e^{(-3-2i)t} (1 + e^{4it})$$

```
Plot[ysol[t], {t, 0, 2}]
```



```
InverseLaplaceTransform[(s + 3) / ((s + 3)^2 + 4), s, t]
1/2 e^(-3-2 i) t (1 + e^4 i t)
```

Here we need to use Euler's formula, $e^{i\theta} = \cos(\theta) + i \sin(\theta)$, invoked by *Mathematica's* ComplexExpand command to offer a "real" presentation of our transform.

```
ComplexExpand[InverseLaplaceTransform[(s + 3) / ((s + 3)^2 + 4), s, t]]
1/2 e^-3 t Cos[2 t] + 1/2 e^-3 t Cos[2 t] Cos[4 t] + 1/2 e^-3 t Sin[2 t] Sin[4 t] +
i (-1/2 e^-3 t Sin[2 t] - 1/2 e^-3 t Cos[4 t] Sin[2 t] + 1/2 e^-3 t Cos[2 t] Sin[4 t])
```

Further, we invoke Mathematica's TrigReduce command to use trigonometric identities to reduce our Inverse Laplace Transform to a simple form from which we could see so much, in this case a damped oscillator!

```
TrigReduce[ComplexExpand[InverseLaplaceTransform[(s + 3) / ((s + 3)^2 + 4), s, t]]]
e^-3 t Cos[2 t]
```

Let us take the Laplace Transform of general functions and derivatives

```
LaplaceTransform[y[t], t, s]
```

```
LaplaceTransform[y[t], t, s]
```

```
LaplaceTransform[y'[t], t, s]
```

```
s LaplaceTransform[y[t], t, s] - y[0]
```

We take Laplace Transform of second derivative . . .

```
LaplaceTransform[y''[t], t, s]
```

```
s^2 LaplaceTransform[y[t], t, s] - s y[0] - y'[0]
```

We take Laplace Transform of third derivative . . .

```
LaplaceTransform[y'''[t], t, s]
```

```
s^3 LaplaceTransform[y[t], t, s] - s^2 y[0] - s y'[0] - y''[0]
```

We take Laplace Transform of fourth derivative . . .

```
LaplaceTransform[y''''[t], t, s]
```

```
s^4 LaplaceTransform[y[t], t, s] - s^3 y[0] - s^2 y'[0] - s y''[0] - y^(3)[0]
```

Let us give $\text{LaplaceTransform}[y[t],t,s]$ a name, say $Y[s]$. Then things look nicer.

$$\text{LaplaceTransform}[y'[t], t, s] /. \{\text{LaplaceTransform}[y[t], t, s] \rightarrow Y[s]\}$$

$$-y[0] + s Y[s]$$

$$\text{LaplaceTransform}[y''[t], t, s] /. \{\text{LaplaceTransform}[y[t], t, s] \rightarrow Y[s]\}$$

$$-s y[0] + s^2 Y[s] - y'[0]$$

Can we take the Laplace Transform of a whole differential equations? Let's try.

$$\text{eq} = \text{LaplaceTransform}[a y''[t] + b y'[t] + c y[t] = \text{Sin}[t], t, s]$$

$$c \text{LaplaceTransform}[y[t], t, s] + b (s \text{LaplaceTransform}[y[t], t, s] - y[0]) +$$

$$a (s^2 \text{LaplaceTransform}[y[t], t, s] - s y[0] - y'[0]) = \frac{1}{1 + s^2}$$

Let us try to replace $\text{LaplaceTransform}[y[t],t,s]$ with $Y[s]$ here too.

$$\text{eq} = \text{LaplaceTransform}[a y''[t] + b y'[t] + c y[t] = \text{Sin}[t], t, s] /.$$

$$\{\text{LaplaceTransform}[y[t], t, s] \rightarrow Y[s]\}$$

$$c Y[s] + b (-y[0] + s Y[s]) + a (-s y[0] + s^2 Y[s] - y'[0]) = \frac{1}{1 + s^2}$$

We can solve for the Laplace of the function in the above differential equation.

$$Y\text{Sol}[s_] = Y[s] /. \text{Solve}[\text{eq}, Y[s]][[1]]$$

$$(1 + b y[0] + a s y[0] + b s^2 y[0] + a s^3 y[0] + a y'[0] + a s^2 y'[0]) / ((1 + s^2) (c + b s + a s^2))$$

Now if we could compute the Inverse Laplace Transform of $Y(s)$ we would have the solution to our differential equation, $y(t)$. Let us do this. Our solution appears a bit on the "wild side" involving the constants a , b , and c , and the all too familiar radical $\sqrt{b^2 - 4ac}$ one obtains from the characteristic equation of the homogeneous portion of our original differential equation $a y''(t) + b y'(t) + c y(t) = \sin(t)$ in using eigenvalue solution strategies.

`InverseLaplaceTransform[YSol[s], s, t] // Simplify`

$$\left(e^{-\frac{(b+\sqrt{b^2-4ac})t}{2a}} \left(-2a^2 - b^2 + 2ac + b\sqrt{b^2-4ac} + 2a^2 e^{\frac{\sqrt{b^2-4ac}t}{a}} + b^2 e^{\frac{\sqrt{b^2-4ac}t}{a}} - \right. \right. \\ \left. \left. 2ac e^{\frac{\sqrt{b^2-4ac}t}{a}} + b\sqrt{b^2-4ac} e^{\frac{\sqrt{b^2-4ac}t}{a}} - 2b\sqrt{b^2-4ac} e^{\frac{(b+\sqrt{b^2-4ac})t}{2a}} \cos[t] - \right. \right. \\ \left. \left. 2(a-c)\sqrt{b^2-4ac} e^{\frac{(b+\sqrt{b^2-4ac})t}{2a}} \sin[t] - a^2 b y[0] - b^3 y[0] + 2abc y[0] - \right. \right. \\ \left. \left. bc^2 y[0] + a^2 \sqrt{b^2-4ac} y[0] + b^2 \sqrt{b^2-4ac} y[0] - 2ac \sqrt{b^2-4ac} y[0] + \right. \right. \\ \left. \left. c^2 \sqrt{b^2-4ac} y[0] + a^2 b e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] + b^3 e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] - 2abc e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] + \right. \right. \\ \left. \left. bc^2 e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] + a^2 \sqrt{b^2-4ac} e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] + b^2 \sqrt{b^2-4ac} e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] - \right. \right. \\ \left. \left. 2ac \sqrt{b^2-4ac} e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] + c^2 \sqrt{b^2-4ac} e^{\frac{\sqrt{b^2-4ac}t}{a}} y[0] - 2a^3 y'[0] - \right. \right. \\ \left. \left. 2ab^2 y'[0] + 4a^2 c y'[0] - 2ac^2 y'[0] + 2a^3 e^{\frac{\sqrt{b^2-4ac}t}{a}} y'[0] + 2ab^2 e^{\frac{\sqrt{b^2-4ac}t}{a}} y'[0] - \right. \right. \\ \left. \left. 4a^2 c e^{\frac{\sqrt{b^2-4ac}t}{a}} y'[0] + 2ac^2 e^{\frac{\sqrt{b^2-4ac}t}{a}} y'[0] \right) \right) / \left(2\sqrt{b^2-4ac} (a^2 + b^2 - 2ac + c^2) \right)$$

The **Transfer function** of a differential equation such as $a*y''(t) + b*y'(t) + c*y(t) = \sin(t)$ is $Q(s) = 1 / (a*s^2 + b*s + c)$ (1 over the denominator in the Laplace Transform of the solution) and notice the coefficients a, b, and c, of the LHS or of the system - not the driver $\sin(t)$. This characterizes the system response and the roots of the characteristic equation, $a*s^2 + b*s + c = 0$, are the characteristic roots or eigenvalues of this system. Engineers get to "know" systems and the systems' behaviors by their Transfer functions and the eigenvalues; electrical engineers call the eigenvalues "poles."

Indeed, if we let $y(0) = 0$ and $y'(0) = 0$ then $YSol[s] = L[s]*Q[s]$, and so $Q[s] = Ysol[s]/L[s]$, where $YSol[s]$ is the output's Laplace Transform and $L[s]$ is the input's Laplace Transform. This effectively shows that the Laplace Transform of the Solution is "simply" a multiplication of Laplace Transform of the input or driver function, i.e. $L[s]$, and the transfer function $Q[s] = Q(s) = 1 / (a*s^2 + b*s + c)$. The latter contains all the "action" of the differential equation model terms $a*y''(t) + b*y'(t) + c*y(t)$ while the former, $L[s]$, contains the "action" of the driver function, in this case $\sin(t)$.

So when we actually use reasonable initial conditions ($y(0) = 0$, no initial displacement, and $y'(0) = 0$, no initial velocity - depending upon the driver to get some activity going) on the inverse Laplace Transform, $YSol[s]$, of our differential equation $a*y''(t) + b*y'(t) + c*y(t) = \sin(t)$

`YSol[s] /. {y[0] -> 0, y'[0] -> 0}`

$$\frac{1}{(1 + s^2) (c + b s + a s^2)}$$

We try a specific example, one we know we can solve in several other ways - by hand, with *Mathematica* `DSolve`. We consider

$$y''(t) + 6 y'(t) + 5 y(t) = \sin(t), \quad y(0) = 4, \quad y'(0) = 0.$$

First we use `DSolve` and some formatting to grab the solution for comparison.

```
ysol[t_] =
  y[t] /. DSolve[{1 y''[t] + 6 y'[t] + 5 y[t] == Sin[t], y[0] == 4, y'[0] == 0}, y[t], t][[
    1]] // Expand
-  $\frac{105}{104} e^{-5t} + \frac{41 e^{-t}}{8} - \frac{3 \text{Cos}[t]}{26} + \frac{\text{Sin}[t]}{13}$ 
```

We consider this same differential equation and apply Laplace Transforms to all terms, i.e. both sides of the differential equation.

```
eq = LaplaceTransform[1 y''[t] + 6 y'[t] + 5 y[t] == Sin[t], t, s] /.
  {LaplaceTransform[y[t], t, s] -> Y[s], y[0] -> 4, y'[0] -> 0}
```

$$-4 s + 5 Y[s] + s^2 Y[s] + 6 (-4 + s Y[s]) = \frac{1}{1 + s^2}$$

Now we solve for $Y[s]$, the Laplace Transform of $y[t]$ in order to get ready for the inverse Laplace Transform.

```
L[s_] = Y[s] /. Solve[eq, Y[s]][[1]]
```

$$\frac{25 + 4 s + 24 s^2 + 4 s^3}{(1 + s^2) (5 + 6 s + s^2)}$$

Now we apply the inverse transform to get the solution for $y[t]$.

```
l[t_] = InverseLaplaceTransform[L[s], s, t] // Expand
```

$$- \frac{105}{104} e^{-5t} + \frac{41 e^{-t}}{8} - \frac{3 \text{Cos}[t]}{26} + \frac{\text{Sin}[t]}{13}$$

Notice the two portions of this solution, namely the Transient solution

$- \frac{105}{104} e^{-5t} + \frac{41 e^{-t}}{8}$ and the Steady State $- \frac{3 \text{Cos}[t]}{26} + \frac{\text{Sin}[t]}{13}$. These would be

produced by various hand techniques or *Mathematica*'s DSolve approach which we show here.

How does that compare to the DSolve solution? Exactly the same as we see when, again, we use DSolve to obtain a solution.

```
ysol[t_] =
  y[t] /. DSolve[{1 y''[t] + 6 y'[t] + 5 y[t] == Sin[t], y[0] == 4, y'[0] == 0}, y[t], t][[
    1]] // Expand
-  $\frac{105}{104} e^{-5t} + \frac{41 e^{-t}}{8} - \frac{3 \text{Cos}[t]}{26} + \frac{\text{Sin}[t]}{13}$ 
```

So what is the big deal? Who would someone choose Laplace Transforms to solve a problem if DSolve or by hand can do it directly?


Here is the point (well one of the many points of advantage) of Laplace Transforms. They convert calculus to algebra, then we play in algebra land, and then we return to calculus land. More formally, the Laplace Transform transforms the problem from the time domain (t) to the frequency domain in (s).

We consider two additional "Real Good" reasons for Laplace Transforms

Unit Step or Heaviside Function - what does this function do.

```
f1[x_] = UnitStep[x]
UnitStep[x]
```


```
Plot[f1[x], {x, -8, 8}, PlotStyle -> {Thickness[.01]}]
```



```
f2[x_] = UnitStep[x - 3]
```

```
UnitStep[-3 + x]
```

```
Plot[f2[x], {x, -8, 8}, PlotStyle -> {Thickness[.01]}]
```




We could define our own unit step function with conditional or if statements as follows, but *Mathematica* already has it defined so we accept `UnitStep[x]` in *Mathematica*.

$\text{unitstep}(x) = 1$, if $x > 0$, and 0 otherwise.

```
unitstep[x_] = If[x > 0, 1, 0]
```

```
If[x > 0, 1, 0]
```

```
Plot[unitstep[x], {x, -8, 8}, PlotStyle -> {Thickness[.01]}]
```




Dirac Delta or Impulse Function

```
g[x_] = DiracDelta[x]
```

```
DiracDelta[x]
```

Strange plot

```
Plot[g[x], {x, -4, 4}, PlotStyle -> {Thickness[.01]}]
```



We examine the Dirac Delta function some more. This function represents an instantaneous change. E.g., we change the salt concentration by "dumping" in more salt ALL AT ONCE! OR we "shock" a spring mass giving it an instantaneous force ALL AT ONCE!

The integral of an interval containing $x = 0$ for the `DiracDelta[x]` function is 1. What this says is the impulse offered by a driver of the form `DiracDelta[x]` imparts force of size 1 unit to our system, BUT all at once like the quick bang of a hammer, rather than a gradual application of force.

We check out some properties of the `DiracDelta[x]` function.

```
Integrate[g[x], {x, -1, 1}]
```

1

```
Integrate[g[x], {x, .5, 1}]
```

```
0.
```

```
g[.1]
```

```
0
```

```
g[0]
```

```
DiracDelta[0]
```

```
g[2]
```

```
0
```

Transforms of UnitStep and Dirac Function - where does this fit into our Table of Laplace transforms?

```
f[t_] = UnitStep[t];
```

```
LaplaceTransform[f[t], t, s]
```

$$\frac{1}{s}$$

```
g[t_] = DiracDelta[t];
```


```
LaplaceTransform[g[t], t, s]
```

```
1
```

UnitStep function represents a "step" or finite immediate change in something. DiracDelta function represents an instantaneous action at one point. . . zero elsewhere.

$$\text{del}[x_, a_] = \frac{1}{2a} (\text{UnitStep}[x - (-a)] - \text{UnitStep}[x - a] - \text{UnitStep}[-a + x] + \text{UnitStep}[a + x])$$

```
Plot[del[x, 1], {x, -8, 8}]
```




```
Integrate[del[x, 1], {x, -100, 100}]
```

1

We examine an animation or a Manipulation of $\text{del}[x,a]$ as a gets smaller and smaller. We say $\text{DiracDelta}[x]$ is the limit of $\text{del}[x,a]$ as $a \rightarrow 0$.

Here are the plots in Manipulation to illustrate this point. By clicking on the + box in this command one can see the values of a and by using the slider next to a one can see the effect of decreasing the value of a in the plot of the resulting function.


```
Manipulate[Plot[del[x, a], {x, -8, 8},
  PlotRange -> {{-8, 8}, {0, 10}}, PlotPoints -> 100], {a, 3, .001, -.0001}]
```



We can combine UnitStep functions to turn things on and then off. Consider this function.

```
h[x_] = UnitStep[x - 3] - UnitStep[x - 10]
-UnitStep[-10 + x] + UnitStep[-3 + x]

Plot[h[x], {x, 0, 20}, PlotStyle -> {Thickness[.01]}
```



Suppose we have a mass spring system and we want to see the effect of adding additional weight (force) on the mass after it is in motion, say adding 5 units of force at time $t = 2$.

```
eq = LaplaceTransform[1 y''[t] + 6 y'[t] + 5 y[t] == 5 UnitStep[t - 2], t, s] /.
  {LaplaceTransform[y[t], t, s] -> Y[s], y[0] -> 4, y'[0] -> 0}
-4 s + 5 Y[s] + s^2 Y[s] + 6 (-4 + s Y[s]) ==  $\frac{5 e^{-2 s}}{s}$ 
```

Now we solve for $Y[s]$, the Laplace Transform of $y[t]$ in order to get ready for the inverse Laplace Transform.


```
L[s_] = Y[s] /. Solve[eq, Y[s]][[1]]
 $\frac{e^{-2 s} (5 + 24 e^{2 s} s + 4 e^{2 s} s^2)}{s (5 + 6 s + s^2)}$ 
```

Now we apply the Inverse Laplace Transform to get the solution for $y[t]$.

```
y1[t_] = InverseLaplaceTransform[L[s], s, t]
 $\frac{1}{4} e^{-5 t} (-4 + 20 e^{4 t} + (e^{10} + 4 e^{5 t} - 5 e^{2+4 t}) \text{HeavisideTheta}[-2 + t])$ 
```

From the plot we see the change in the action at $t = 2$ due to our applying additional force at that time.

```
Plot[y1[t], {t, 0, 10}, PlotRange -> {{0, 10}, {0, 4}}]
```



Suppose we have a spring mass system and we want to see the effect of hitting it with spike force of 4 units of force after two ($t = 2$) seconds of "usual" motion.

```
eq = LaplaceTransform[1 y''[t] + 6 y'[t] + 5 y[t] == 4 DiracDelta[t - 2], t, s] /.
  {LaplaceTransform[y[t], t, s] -> Y[s], y[0] -> 0, y'[0] -> 0}
5 Y[s] + 6 s Y[s] + s^2 Y[s] == 4 e^{-2 s}
```

Now we solve for $Y[s]$, the Laplace Transform of $y[t]$ in order to get ready for the inverse Laplace Transform.

```
L[s_] = Y[s] /. Solve[eq, Y[s]][[1]]

$$\frac{4 e^{-2 s}}{5 + 6 s + s^2}$$

```

Now we apply the transform to get the solution for $y[t]$.


```
y1[t_] = InverseLaplaceTransform[L[s], s, t]

$$e^{-5(-2+t)} (-1 + e^{4(-2+t)}) \text{HeavisideTheta}[-2+t]$$

```

From the plot we see the change in the action at $t = 2$ due to our applying a spike force at that time. Up until the spring mass was at rest due to its initial conditions of $y(0) = 0$ and $y'(0) = 0$.


```
Plot[y1[t], {t, 0, 10}, PlotRange -> {{0, 10}, {0, 1}}]
```



Let us consider a tank system in which we "dump" stuff (DiracDelta function) and we alter amounts suddenly (UnitStep function). First the system without any drastic dumping. Can you draw a scenario for this system given the equations?

$$\text{eq} = \{x'[t] == (4 * y[t]) / 200 - (7 * x[t]) / 200,$$

$$y'[t] == (7 * x[t]) / 200 - (7 * y[t]) / 200, x[0] == 50, y[0] == 0\}$$


$$\{x'[t] == -\frac{7x[t]}{200} + \frac{y[t]}{50}, y'[t] == \frac{7x[t]}{200} - \frac{7y[t]}{200}, x[0] == 50, y[0] == 0\}$$

```
sol = DSolve[eq, {x[t], y[t]}, t]
```

$$\left\{ \left\{ x[t] \rightarrow 25 e^{-\frac{7t}{200} - \frac{\sqrt{7}t}{100}} \left(1 + e^{\frac{\sqrt{7}t}{50}} \right), y[t] \rightarrow \frac{25}{2} \sqrt{7} e^{-\frac{7t}{200} - \frac{\sqrt{7}t}{100}} \left(-1 + e^{\frac{\sqrt{7}t}{50}} \right) \right\} \right\}$$

```
xs[t_] = x[t] /. sol[[1]]; ys[t_] = y[t] /. sol[[1]];
```

```
Plot[{xs[t], ys[t]}, {t, 0, 100}, PlotRange -> {{0, 100}, {0, 50}}]
```



Let us consider a tank system in which we (I) "dump" stuff (DiracDelta function),

(2) we alter amounts suddenly (UnitStep) function, or combinations of these. What are we dumping into the system? Where are we dumping it in? When do we start dumping? How often are we dumping it in?

What does these stuff functions do?

```
stuff1[t_] = 40 DiracDelta[t - 25];
stuff2[t_] = 40 UnitStep[t - 25];
stuff3[t_] = 40 UnitStep[t - 25] - 20 UnitStep[t - 30];
stuff4[t_] = Sum[20 DiracDelta[t - a], {a, 20, 200, 20}];
```

We might then modify the model system with the new "stuff" function and Laplace away!

```
stuff[t_] = stuff1[t]
40 DiracDelta[-25 + t]
```

```
eq = {x'[t] == 4 y[t] / 200 - 7 x[t] / 200 + stuff[t],
      y'[t] == 7 x[t] / 200 - 7 y[t] / 200,
      x[0] == 50, y[0] == 0}
```

$$\left\{ \begin{aligned} x'[t] &= 40 \text{DiracDelta}[-25 + t] - \frac{7x[t]}{200} + \frac{y[t]}{50}, \\ y'[t] &= \frac{7x[t]}{200} - \frac{7y[t]}{200}, \quad x[0] = 50, y[0] = 0 \end{aligned} \right.$$

```
eq[[1]]
```

$$x'[t] = 40 \text{DiracDelta}[-25 + t] - \frac{7x[t]}{200} + \frac{y[t]}{50}$$

Now use LaplaceTransform syntax, solve, plot, and examine solution.

```
eq1 = LaplaceTransform[eq[[1]], t, s] /.
```

```
{LaplaceTransform[x[t], t, s] -> X[s], LaplaceTransform[y[t], t, s] -> Y[s],
 x[0] -> 50, y[0] -> 0}
```

$$-50 + s X[s] = 40 e^{-25s} - \frac{7X[s]}{200} + \frac{Y[s]}{50}$$

```
eq2 = LaplaceTransform[eq[[2]], t, s] /.
```

```
{LaplaceTransform[x[t], t, s] -> X[s], LaplaceTransform[y[t], t, s] -> Y[s],
 x[0] -> 50, y[0] -> 0}
```

$$s Y[s] = \frac{7X[s]}{200} - \frac{7Y[s]}{200}$$

XYsol = Solve[{eq1, eq2}, {X[s], Y[s]}]

$$\left\{ \left\{ X[s] \rightarrow \frac{2000 e^{-25 s} (4 + 5 e^{25 s}) (7 + 200 s)}{21 + 2800 s + 40000 s^2}, Y[s] \rightarrow \frac{14000 e^{-25 s} (4 + 5 e^{25 s})}{21 + 2800 s + 40000 s^2} \right\} \right\}$$

Xsol[s_] = X[s] /. XYsol[[1]] // Expand

$$\frac{70000}{21 + 2800 s + 40000 s^2} + \frac{56000 e^{-25 s}}{21 + 2800 s + 40000 s^2} + \frac{2000000 s}{21 + 2800 s + 40000 s^2} + \frac{1600000 e^{-25 s} s}{21 + 2800 s + 40000 s^2}$$

Ysol[s_] = Y[s] /. XYsol[[1]] // Expand

$$\frac{70000}{21 + 2800 s + 40000 s^2} + \frac{56000 e^{-25 s}}{21 + 2800 s + 40000 s^2}$$

xsol[t_] = InverseLaplaceTransform[Xsol[s], s, t]


$$5 \left(5 e^{-\frac{1}{200} (7+2\sqrt{7}) t} \left(1 + e^{\frac{\sqrt{7} t}{50}} \right) + 4 e^{-\frac{1}{200} (7+2\sqrt{7}) (-25+t)} \left(1 + e^{\frac{1}{50} \sqrt{7} (-25+t)} \right) \right) \text{HeavisideTheta}[-25 + t]$$

ysol[t_] = InverseLaplaceTransform[Ysol[s], s, t]

$$-\frac{5}{2} \sqrt{7} \left(-5 e^{-\frac{1}{200} (7+2\sqrt{7}) t} \left(-1 + e^{\frac{\sqrt{7} t}{50}} \right) - 4 e^{-\frac{1}{200} (7+2\sqrt{7}) (-25+t)} \left(-1 + e^{\frac{1}{50} \sqrt{7} (-25+t)} \right) \right) \text{HeavisideTheta}[-25 + t]$$

After solving using Laplace Transform techniques we plot our solution for x(t). We see that from the dumping due to $\text{stuff}[t] = 40 \cdot \text{DiracDelta}[t-25]$ of 40 additional units instantly at time $t = 25$ there is a jump in x(t) at time $t = 25$ and an attendant rapid change in the second tank, y(t).


xPlot = Plot[xsol[t], {t, 0, 200}, PlotRange -> {{0, 200}, {0, 100}}]



```

yPlot =
  Plot[ysol[t], {t, 0, 200}, PlotRange -> {{0, 200}, {0, 100}}, PlotStyle -> Hue[.3]]

```




Here we see both $x(t)$ in gold and $y(t)$ in green showing the effect of our dump in both tanks.

```

Show[xPlot, yPlot]


```



Suppose we have a spring mass system and we want to see the effect of turning on a force and turning off that force, over and over again, perhaps by placing the mass in an alternating magnetic field. We model this with small rectangular functions of forces, 0.5 units high and 1 unit long, spaced 2 units apart, centered at 1, 3, 5, etc.

```
rut[t_] = Sum[5 UnitStep[t - 2 k] - 5 UnitStep[t - (2 k + 1)], {k, 0, 10}]
- 5 UnitStep[-21 + t] + 5 UnitStep[-20 + t] - 5 UnitStep[-19 + t] +
  5 UnitStep[-18 + t] - 5 UnitStep[-17 + t] + 5 UnitStep[-16 + t] -
  5 UnitStep[-15 + t] + 5 UnitStep[-14 + t] - 5 UnitStep[-13 + t] + 5 UnitStep[-12 + t] -
  5 UnitStep[-11 + t] + 5 UnitStep[-10 + t] - 5 UnitStep[-9 + t] + 5 UnitStep[-8 + t] -
  5 UnitStep[-7 + t] + 5 UnitStep[-6 + t] - 5 UnitStep[-5 + t] + 5 UnitStep[-4 + t] -
  5 UnitStep[-3 + t] + 5 UnitStep[-2 + t] - 5 UnitStep[-1 + t] + 5 UnitStep[t]
```

```
pRut = Plot[rut[t], {t, 0, 20}, PlotRange -> {{0, 20}, {0, 6}}
```



We apply our force function $rut(t)$.

```
eq = LaplaceTransform[1 y''[t] + 6 y'[t] + 5 y[t] == rut[t], t, s] /.
  {LaplaceTransform[y[t], t, s] -> Y[s], y[0] -> 0, y'[0] -> 0}
```

$$5 Y[s] + 6 s Y[s] + s^2 Y[s] == \frac{5}{s} - \frac{5 e^{-21 s}}{s} + \frac{5 e^{-20 s}}{s} - \frac{5 e^{-19 s}}{s} + \frac{5 e^{-18 s}}{s} - \frac{5 e^{-17 s}}{s} + \frac{5 e^{-16 s}}{s} - \frac{5 e^{-15 s}}{s} + \frac{5 e^{-14 s}}{s} - \frac{5 e^{-13 s}}{s} + \frac{5 e^{-12 s}}{s} - \frac{5 e^{-11 s}}{s} + \frac{5 e^{-10 s}}{s} - \frac{5 e^{-9 s}}{s} + \frac{5 e^{-8 s}}{s} - \frac{5 e^{-7 s}}{s} + \frac{5 e^{-6 s}}{s} - \frac{5 e^{-5 s}}{s} + \frac{5 e^{-4 s}}{s} - \frac{5 e^{-3 s}}{s} + \frac{5 e^{-2 s}}{s} - \frac{5 e^{-s}}{s}$$

Now we solve for $Y[s]$, the Laplace Transform of $y[t]$ in order to get ready for the inverse Laplace Transform.

L[s_] = Y[s] /. Solve[eq, Y[s]][[1]]

$$\frac{1}{s(5 + 6s + s^2)} \left(5e^{-21s} (-1 + e^s - e^{2s} + e^{3s} - e^{4s} + e^{5s} - e^{6s} + e^{7s} - e^{8s} + e^{9s} - e^{10s} + e^{11s} - e^{12s} + e^{13s} - e^{14s} + e^{15s} - e^{16s} + e^{17s} - e^{18s} + e^{19s} - e^{20s} + e^{21s}) \right)$$


We finally apply the Inverse Laplace Transform to get the solution for y[t].

y1[t_] = InverseLaplaceTransform[L[s], s, t]

$$\begin{aligned} & \frac{1}{4} e^{-5t} \left(1 - 5e^{4t} + 4e^{5t} - (e^{105} + 4e^{5t} - 5e^{21+4t}) \text{HeavisideTheta}[-21+t] + \right. \\ & \left(e^{100} + 4e^{5t} - 5e^{4(5+t)} \right) \text{HeavisideTheta}[-20+t] - e^{95} \text{HeavisideTheta}[-19+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-19+t] + 5e^{19+4t} \text{HeavisideTheta}[-19+t] + \\ & e^{90} \text{HeavisideTheta}[-18+t] + 4e^{5t} \text{HeavisideTheta}[-18+t] - \\ & 5e^{18+4t} \text{HeavisideTheta}[-18+t] - e^{85} \text{HeavisideTheta}[-17+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-17+t] + 5e^{17+4t} \text{HeavisideTheta}[-17+t] + \\ & e^{80} \text{HeavisideTheta}[-16+t] + 4e^{5t} \text{HeavisideTheta}[-16+t] - \\ & 5e^{4(4+t)} \text{HeavisideTheta}[-16+t] - e^{75} \text{HeavisideTheta}[-15+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-15+t] + 5e^{15+4t} \text{HeavisideTheta}[-15+t] + \\ & e^{70} \text{HeavisideTheta}[-14+t] + 4e^{5t} \text{HeavisideTheta}[-14+t] - \\ & 5e^{14+4t} \text{HeavisideTheta}[-14+t] - e^{65} \text{HeavisideTheta}[-13+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-13+t] + 5e^{13+4t} \text{HeavisideTheta}[-13+t] + \\ & e^{60} \text{HeavisideTheta}[-12+t] + 4e^{5t} \text{HeavisideTheta}[-12+t] - \\ & 5e^{4(3+t)} \text{HeavisideTheta}[-12+t] - e^{55} \text{HeavisideTheta}[-11+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-11+t] + 5e^{11+4t} \text{HeavisideTheta}[-11+t] + \\ & e^{50} \text{HeavisideTheta}[-10+t] + 4e^{5t} \text{HeavisideTheta}[-10+t] - \\ & 5e^{10+4t} \text{HeavisideTheta}[-10+t] - e^{45} \text{HeavisideTheta}[-9+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-9+t] + 5e^{9+4t} \text{HeavisideTheta}[-9+t] + \\ & e^{40} \text{HeavisideTheta}[-8+t] + 4e^{5t} \text{HeavisideTheta}[-8+t] - \\ & 5e^{8+4t} \text{HeavisideTheta}[-8+t] - e^{35} \text{HeavisideTheta}[-7+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-7+t] + 5e^{7+4t} \text{HeavisideTheta}[-7+t] + \\ & e^{30} \text{HeavisideTheta}[-6+t] + 4e^{5t} \text{HeavisideTheta}[-6+t] - \\ & 5e^{6+4t} \text{HeavisideTheta}[-6+t] - e^{25} \text{HeavisideTheta}[-5+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-5+t] + 5e^{5+4t} \text{HeavisideTheta}[-5+t] + \\ & e^{20} \text{HeavisideTheta}[-4+t] + 4e^{5t} \text{HeavisideTheta}[-4+t] - \\ & 5e^{4+4t} \text{HeavisideTheta}[-4+t] - e^{15} \text{HeavisideTheta}[-3+t] - \\ & 4e^{5t} \text{HeavisideTheta}[-3+t] + 5e^{3+4t} \text{HeavisideTheta}[-3+t] + \\ & e^{10} \text{HeavisideTheta}[-2+t] + 4e^{5t} \text{HeavisideTheta}[-2+t] - \\ & 5e^{2+4t} \text{HeavisideTheta}[-2+t] - e^5 \text{HeavisideTheta}[-1+t] - \\ & \left. 4e^{5t} \text{HeavisideTheta}[-1+t] + 5e^{1+4t} \text{HeavisideTheta}[-1+t] \right) \end{aligned}$$


Here is our solution plot. Notice the effects of turning on and turning off the driver force.

```
pSol = Plot[y1[t], {t, 0, 20}, PlotRange -> {{0, 20}, {0, 2}}]
```




We examine the input function (square wave) and the output function (rounded wave).

```
Show[pRut, pSol]
```



Up close. Notice the delay of the system (system response) in responding to the turn on and to the shut off of the input function

```
Show[pRut, pSol, PlotRange -> {{0, 5.5}, {0, 6}}]
```



There are many uses and reasons for using Laplace Transforms, indeed, often

engineers live in the frequency domain of the transform world, i.e. s , without returning to the time domain world of t , for they learn to "see" and understand in the frequency world and, as we saw, it is algebra and not calculus - thus making it more "elementary."